HEWLETT PACKARD

Application Note 369-6

Impedance Testing Using Scanner

- HP 4284A Precision LCR Meter -



INTRODUCTION

Today component manufacturers need higher productivity and component yields in order to supply lower cost products with faster delivery and higher reliability to compete against very tough competition. Consequently, most component manufactures require impedance testing using a scanner to improve productivity and to control the quality of their products. But many systems were designed without considering the influence of residual errors specific to scanning systems. The HP 4284A Precision LCR Meter with the Option 301 Scanner Interface has a correction function for a scanning system. This application note describes solutions to the problem of scanning system residual error by comparing a conventional system to an improved system using the HP 4284A. Also this note covers how to design an ideal scanning system.

SCANNING MEASUREMENT PROBLEMS AND SOLUTIONS

This section describes problems, solutions and typical results of using a scanning system to measure components. Part one of this section shows the results of measuring 100 capacitors using the experimental scanning system shown in figure 1. The results show some of the problems of using a scanning system with a conventional LCR meter. Part two shows the results of measuring a capacitor at each channel to analyze what causes the problems, the results makes clear two problems of a conventional systems. And part three shows a solution using the HP 4284A's correction function for a scanning system. Then a practical example of measuring 100 samples is described.



Figure 1. Experimental Scanning System

Problems Using a Conventional Scanning System

The measurement results obtained using a scanner does not always correspond with the values obtained without a scanner. Figure 2 shows the measurement results of measuring 100 capacitors (typical value is 3300 pf) at a measurement frequency of 1 MHz and a oscillator level of 1 V. The scanner system is an experimental system which uses OPEN/SHORT correction and a conventional scanner. The results looks inaccurate because the distribution is wider and some dissipation factor values are negative. Figure 3 shows the measurement results when using the HP 4284A with the HP 16047C Multipurpose Test Fixture to measure the same 100 capacitors. The results shown in figure 3 can be roughly regarded as the actual distribution of the 100 samples. A comparison between them suggests that using a scanner causes this difference. The distribution in measurement values shown in figure 2 is greater than the one shown in figure 3 and some of the D measurement



(a) Capacitance



(b) Dissipation Factor

Figure 2. Measurement Results Using OPEN/SHORT Correction



(a) Capacitance



(b) Dissipation Factor

Figure 3. Measurement Results Using the HP 16047C

values in figure 2 are negative. It seems that the distribution in the measurement value using a scanner is greater than the actual value and that the dissipation factor values tend to be offset to a negative value. The scanner located between fixture and LCR meter has some errors that a conventional LCR meter cannot reduce.

Tracking Down the Error Source

Figure 4 shows the measurement results using a six channel scanner to measure the same capacitor at each channel. The measurement is made with a conventional OPEN/SHORT correction, where the correction measurement data obtained at one of the six channels. The center black dot in figure 4 is the reference value measured using the HP 16047C Multipurpose Test Fixture connected directly to the HP 4284A. The reference capacitance value is 3210 pF and the dissipation factor is 0.00045. Figure 4 clearly shows the two problems of a scanning system using the conventional correction function.

One of problems is the discrepancy in measurement value between each channel. The OPEN/SHORT correction function can reduce errors caused by residuals and stray capacitance of the scanning system. However the OPEN admittance and SHORT impedance of each channel are different from channel to channel because the cable length and the arrangement of parts between the LCR meter and the fixture are different for each channel branch. The reason why this discrepancy occurs is that a conventional LCR meter can store only one set of data for correction.

The other problem is the negative dissipation factor. Figure 4 shows that some scanner measurement values are offset from the reference value to a negative value even if the OPEN/SHORT correction is used. The OPEN/SHORT correction cannot reduce this error because the error is caused by a complicated phase shift of the test signal in the scanner.



Figure 4. Problems Using a Scanner



(a) MULTI-Correction Mode



(b) OPEN/SHORT/LOAD Correction in MULTI-Correction Mode

Figure 5. Two Solutions offered by the HP 4284A

Figure 5 shows measurement results when the HP 4284A's MULTI-correction mode and OPEN/SHORT/LOAD correction are used. A detailed discussion is given in the next.

MULTI-Correction Mode

When the Option 301 Scanner Interface is installed the HP 4284A can minimize the discrepancy in measurement values between the scanner measurement channels with it's MULTI-correction mode function. With this option, the HP 4284A can store a complete set of correction data (OPEN/SHORT/LOAD) for up to 128 channels. All the data is stored in the HP 4284A 's non-volatile memory, therefore it is not necessary to transfer data from the controller to the HP 4284A (the data is not lost when the HP 4284A is turned off). Figure 5-(a) shows the measurement results when using OPEN/SHORT correction in MULTI-correction mode, less measurement discrepancy.

OPEN/SHORT/LOAD Correction

The discrepancy in measurement data can be minimized by using the MULTI-correction function, but the negative dissipation factor problem is still not solved. The OPEN/SHORT/LOAD correction functions built into the HP 4284A will solve this problem. Figure 5-(b) shows the measurement results when using OPEN/SHORT/LOAD correction in the MULTI-correction mode. The results show that the error in dissipation factor is reduced and that all measurement values are close to the reference value.

The OPEN/SHORT/LOAD correction requires OPEN, SHORT, and working standards. The working standard should have good stability and its characteristics should be accurately known and close to the value to be measured. To measure the characteristics of a working standard, use the HP 4284A with a multipurpose test fixture such as the HP 16047A/C.

Confirmation Measurement

Figure 6 shows the confirmation measurement results for the same 100 capacitors used before and with the OPEN/SHORT/LOAD and MULTI-correction mode. The measurement results shown in figure 6 closely matches those in figure 3. This demonstrates that these correction functions are effective in reducing errors due to the scanning system.

DESIGNING A SCANNING SYSTEM

This section describes how to design a scanning system to measure components using an LCR meter.

Reviewing Specifications

When you design a scanning system, you have to select the scanner and the cable configuration. To do this, you should clarify your measurement requirements (specifications) for the system. For example, it is impossible to measure a 10 pF capacitor with 0.1% accuracy when using a scanner that has 100 pF of stray capacitance. Specifications to review are measurement accuracy, the DUT measurement frequency, and the number of scanner channels. The measurement accuracy requirement will determine the maximum residual and stray capacitance allowed in the scanning system. The measurement value and the test frequency requirements are determined by the cable configuration. The number of channels determines the type of scanner you should select.



(a) Capacitance



(b) Dissipation Factor

Figure 6. Improved Measurement Results

Consideration for Residuals

The OPEN/SHORT/LOAD compensation function of the HP 4284A can reduce most of the errors due to the scanning system. It, however, cannot zero out the errors if the residuals in the system are too great. Residuals consist of series impedance and parallel admittance that exist in the scanner and the interconnecting cables up to the DUT. To find these residual values, refer to the data sheets of the scanner and cables used, or directly measure them by measuring with the DUT terminals OPEN and SHORTed.

The series impedance and parallel admittance values should be less than 1/100 of the DUT's value. For example, when a 100 pF capacitor is measured at 1 MHz, the series impedance should be less than 16 ohms (2.5 micro Henrys in inductance) and the parallel admittance should be less than 6.3 micro Siemens (1 pF in capacitance). If the residuals in your system exceed (or are estimated to exceed) the limits, try minimizing the residuals by adopting a 4-terminal configuration, changing the wires to shielded cables, and guarding the test fixture.

Selecting Cable Configuration and Scanner

The cable configuration used for a scanner depends on the impedance value to be measured. There are five cable configurations used to measure impedance as shown in Figure 7. Figure 7 shows the appropriate cable configurations for the impedance ranges to be measured. The 3-terminal, 5-terminal, or 4-terminal pair are recommended for high impedance measurements, and the 4-terminal or 4-terminal pair are recommended for low impedance measurements. But if the impedance of a test device is in the mid-range impedance values, the 2-terminal configuration can be used. Using the 2-terminal configuration lets you have the maximum number of measurement channels with the minimum number of instruments, thus lower system cost. Using the 4-terminal pair configuration will achieve the highest accuracy over a wide impedance measurement range at a higher system cost.

After you determine the cable configuration you can select a scanner. As table 1 shows, the maximum number of channels depends the cable configuration and the scanner used. Using Hewlett Packard's scanners (for example, the HP 3488A or HP 3235A) makes the design of a system easy because they are controlled via HP-IB, plus many kinds of switching modules are provided.

Table 1. Channels Available vs Cable and Scanner Configuration

Recommanded Impedance Range	Middle Impedance	Very high Impedance	Low Impedance	Low to high Impedance	Very low to high Impedance
Scanner \ Cable Config.	2 terminal	3 terminal	4 terminal	5 terminal	4 terminal-pair
3488A with opt. 010	10ch 46ch* 91ch**	—	10ch 100ch***	—	—
3488A with opt. 012	—	4ch 16ch*	—	4ch	—
3235A + 34504A	—	6ch 36ch*	—	6ch 36ch**	6ch 36ch**

* connecting in series
** connecting in series and using 2 scanners
*** connecting in series and using 5 scanners



(a) 2 Terminal



(b) 3 Terminal



(c) 4 Terminal



(d) 5 Terminal



(e) 4 Terminal-Pair

Note:

Only the combination of the HP 3235A with the HP 34504A can provide a 4-terminal-pair configuration. To configure a 4-terminal-pair, both the inner and outer conductors must be switched ON and OFF.

Figure 7. Cable Configurations

Programming

After the hardware specifications of the system are fixed, you can write a control program to take measurement and correction data. The HP 4284A requires a controller to use the MULTI-correction mode. Program 1 lists a sample program. This program controls the HP 4284A with Option 301, and the HP 3488A with Option 010, and scans six channels when using the 2-terminal configuration. You can modify this program to suit your measurement system and scanner needs. The following procedure describes how to use this sample program.



Figure 8. Flow Chart of Sample Program

If you apply a high DC or AC current to scanner, the measuring sequence should be modified as follows.

(1) Turn off the test signal or bias
(2) Switch channels
(3) Apply the test signal or bias
(4) Measure

This program scans six channels to make correction data measurements. While the program is running a message will be displayed on the controller's screen.

(STEP 1) Select function

The following message is displayed on the screen when the program starts.

SELECT FUNCTION
(1) MEASUREMENT
(2) CORRECTION ?

This is asking if you want to take correction data or make a measurement. If you select to take correction data, you should type 1 and press the controller's **RETURN** key. If your select MEASUREMENT, then you should press 2 and press the **RETURN** key.

(STEP 2) Select correction method

After pressing **RETURN** key in step (1), the following message is displayed.

SELECT CORRECTION METHOD TO BE USED IN MEASUREMENT
(1) OPEN/SHORT CORRECTION
(2) OPEN/SHORT/LOAD CORRECTION

Select the correction method to be used in the measurement. You should select 1 or 2 and press the **RETURN** key.

(STEP 3) Selection correction mode to be used for measurement

After the **RETURN** key is pressed the following message will be displayed:

SELECT CORRECTION MODE
(1) SINGLE (2) MULTI

You should select 1 or 2 and press the **RETURN** key. The next step depends on the selection of STEP 1 (getting correction data or measuring device).
If you chose to measure, the system starts scanning and measuring the devices. If you chose to get correction data, the program proceeds to STEP **4**.

(STEP 4) Measure the OPEN correction data for each channel

The following message will be displayed on the screen:

CH1 OPEN MEASUREMENT
OPEN TEST TERMINALS OF CH1,
THEN PRESS CONTINUE KEY

After this message is displayed, you should open the test terminal for channel 1 and press the **CONTINUE** softkey of the controller to measure the open admittance. When you press the CONTINUE key, the HP 4284A starts measuring the open admittance of channel 1 and displays the following message:

OPEN MEASUREMENT IN PROGRESS

If you select the MULTI-correction mode, you should repeat the same procedure for channels 2 to 6.

After the measurement is completed the next message is displayed.

(STEP 5) Measure SHORT correction data for each channel

CH.1 SHORT MEASUREMENT
SHORT TEST TERMINALS OF CH1,
THEN PRESS CONTINUE KEY

After this message is displayed, you should short the test terminal of channel 1 and press the **CONTINUE** softkey of the controller to measure the short impedance. When you press the **CONTINUE** key, the HP 4284A starts measuring the short impedance of channel 1 and displays the following message.

SHORT MEASUREMENT IN PROGRESS

If you select MULTI-correction mode, you should repeat the same procedure for channels 2 to 6.

After measuring, if you chose the OPEN/SHORT/LOAD correction, the program will proceed to STEP 6. If you did not select OPEN/SHORT/LOAD correction, the program will proceed to STEP 7.

(STEP 6) Measure LOAD correction data for each channel

CH.1 LOAD MEASUREMENT
CONNECT A STANDARD TO TEST
TERMINALS OF CH1, THEN PRESS CONTINUE
KEY

After this message is displayed, connect the standard device to the test terminals of channel 1 and press **CONTINUE** key. The HP 4284A will start to measure the device and display the following message.

LOAD MEASUREMENT IN PROGRESS

After the measurement is complete, the following message is displayed.

CH.1 STANDARD VALUE SETTING
INPUT STANDARD VALUE OF A

Type in the standard's value of A (A means first parameter)

INPUT STANDARD VALUE OF B

Type in the standard's value of B (B means the second parameter and press the **RETURN** key. The following message will be displayed.

STANDARD VALUE
A= xxxxxx, B= xxxxxx, OK?
(1) YES (2) NO

If the value is wrong, type 2 and press the **RETURN** key. You can re-enter the standard value (xxxxxx means the standard value that you entered). If you chose YES (type 1 and press the **RETURN** key), getting the correction data for channel 1 finishes. If you select the MULTI-correction mode, you should repeat the same procedure for channels 2 to 6. After getting all the correction data, the following message will be displayed.

CORRECTION IS COMPLETED.

DO YOU WANT TO CONTINUE TO MEASURE ?
(1) YES (2) NO

To start scanning, select YES (type 1 and press the **RETURN** key). To exit the program, select NO (type 2 and press the **RETURN** key).

(STEP 7) Scanning measurement

This program scans six channels to measure and display the measurement data on the screen.

Typical Measurement Results

Figure 9 shows sample results of a scanning measurement using the HP 4284A and the HP 3488A. All measurements in figure 9 use the OPEN/SHORT/LOAD correction and the MULTI-correction mode. The sample devices are capacitors and inductors. Figure 9-(a) shows the results of measuring high value capacitors, figure 9-(b) shows the results of measuring low value capacitors and figure 9-(c) shows the results of inductance measurement. The results shows that the measurement results are not influenced by the scanner, hence the OPEN/SHORT/LOAD correction and the MULTI-correction mode are effective for all impedance ranges.

CONCLUSION

The HP 4284A with Option 301 has powerful correction functions for reducing the errors that a conventional LCR meter can not reduce when making scanning measurements. The HP 4284A is an outstanding LCR meter to use for obtaining accurate scanning system impedance measurement results.

	CAPACITANCE [uF]	DISSIPATION FACTOR
REFERENCE VALUE	910.870	0.09964
CHANNEL 1	909.658	0.05706
CHANNEL 2	909.702	0.05771
CHANNEL 3	909.659	0.05736
CHANNEL 4	909.463	0.05861
CHANNEL 5	909.464	0.05754
CHANNEL 6	910.748	0.05838

(Test Frequency 100Hz Osc. level 1V)

(a) High Capacitance

	CAPACITANCE [pF]	DISSIPATION FACTOR
REFERENCE VALUE	9.60644	0.001511
CHANNEL 1	9.63423	0.001572
CHANNEL 2	9.64935	0.001467
CHANNEL 3	9.63341	0.001628
CHANNEL 4	9.69686	0.002124
CHANNEL 5	9.60916	0.001511
CHANNEL 6	9.61122	0.002082

(Test frequency 1MHz Osc. level 1V)

(b) Low Capacitance

	INDUCTANCE [uH]	DISSIPATION FACTOR
REFERENCE VALUE	91.1340	6.05477
CHANNEL 1	90.9654	6.03313
CHANNEL 2	91.3405	6.06925
CHANNEL 3	91.2980	6.05693
CHANNEL 4	91.2880	6.06066
CHANNEL 5	91.2962	6.06082
CHANNEL 6	91.2363	6.06563

(Test frequency 1kHz Osc. level 10mA)

(c) Inductance

Figure 9. Measurement Results

Program 1. Sample Program

```
1000  !***********************************
1010  !* 4284A APPLICATION NOTE 369-6    *
1020  !* Impedance Testing Using Scanner *
1030  !***********************************
1040  DIM A(1:100),B(1:100),Work$[100]
1050  DIM Open_ch$(1:100)[15],Close_ch$(1:100)[15],Open_all_ch$[27]
1060                                          ! <<Setting constants>>
1070  Hp4284a=717                             ! HP 4284A's address
1080  Hp3488a=709                             ! HP 3488A's address
1090  Freq=1.E+6                              ! Frequency is 1MHz
1100  Meas_function$="CPD"                    ! Meas. function is Cp-D
1110  Corr_function$=Meas_function$           ! Load corr. function is CP-D
1120  Noch=6                                  ! Number of channel
1130  A$=Meas_function$[1,2]                  !
1140  B$=Meas_function$[3,3]                  !
1150                                          !
1160  Open_ch$(1)="OPEN101"                   ! Command to open channel 1
1170  Open_ch$(2)="OPEN102"                   ! Command to open channel 2
1180  Open_ch$(3)="OPEN103"                   ! Command to open channel 3
1190  Open_ch$(4)="OPEN104"                   ! Command to open channel 4
1200  Open_ch$(5)="OPEN105"                   ! Command to open channel 5
1210  Open_ch$(6)="OPEN106"                   ! Command to open channel 6
1220  Close_ch$(1)="CLOSE101"                 ! Command to close channel 1
1230  Close_ch$(2)="CLOSE102"                 ! Command to close channel 2
1240  Close_ch$(3)="CLOSE103"                 ! Command to close channel 3
1250  Close_ch$(4)="CLOSE104"                 ! Command to close channel 4
1260  Close_ch$(5)="CLOSE105"                 ! Command to close channel 5
1270  Close_ch$(6)="CLOSE106"                 ! Command to close channel 6
1280  Open_all_ch$="OPEN101,102,103,104,105,106"! Command to close all channel
1290                                          !
1300  Measurement=0                           !
1310  Correction=1                            !
1320  Open_short=0                            !
1330  Open_short_load=1                       !
1340  Single=0                                !
1350  Multi=1                                 !
1360                                          !<<Setting initial value>>
1370  A_sum=0                                 ! Summation of A
1380  B_sum=0                                 ! Summation of B
1390  A_sx=0                                  ! S.D. of A
1400  B_sx=0                                  ! S.D. of B
1410                                          !<<Setting condition>>
1420  Work=0                                  !<<Selecting function>>
1430  PRINT CHR$(12)                          ! Clear screen
1440  PRINT "SELECT FUNCTION (1) MEASUREMENT   (2) CORRECTION ? "
1450  INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
1460  IF Work=1 THEN Function=Measurement     !
1470  IF Work=2 THEN Function=Correction      !
1480  IF Function=Correction THEN 1500        !
1490                                          !<<Selecting correction>>
1500  Work=0                                  !
1510  PRINT CHR$(12)                          ! Clear screen
1520  PRINT "SELECT CORRECTION METHOD TO BE USED IN MEASUREMENT "
1530  PRINT "(1) OPEN/SHORT  (2) OPEN/SHORT/LOAD"
1540  INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
1550  IF Work=1 THEN Correct_sw=Open_short    !
1560  IF Work=2 THEN Correct_sw=Open_short_load !
1570  IF Work<>1 AND Work<>2 THEN 1520        !
1580                                          !<<Selection correction mode>>
1590  Work=0                                  !
1600  PRINT CHR$(12)                          ! Clear screen
1610  PRINT "SELECT CORRECTION MODE  (1) SINGLE   (2) MULTI "
1620  INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
1630  IF Work=1 THEN Correct_mode=Single      !
1640  IF Work=2 THEN Correct_mode=Multi       !
1650  IF Work<>1 AND Work<>2 THEN 1620        !
1660                                          !<<Setting string>>
1670  IF Correct_sw=Open_short THEN Correct_sw$="OPEN/SHORT"
1680  IF Correct_sw=Open_short_load THEN Correct_sw$="OPEN/SHORT/LOAD"
1690  IF Correct_mode=0 THEN Correct_mode$="SING"
1700  IF Correct_mode=1 THEN Correct_mode$="MULT"
1710                                          !<<Initializing instrument>>
1720  OUTPUT Hp4284a;"*RST"                   ! Reset HP4284A
1730  OUTPUT Hp3488a;Open_all_ch$             ! Open all channel of HP3488A
1740  OUTPUT Hp4284a;"FREQ "&VAL$(Freq)       ! MEASUREMENT FREQUENCY
1750  OUTPUT Hp4284a;"FUNC:IMP "&Meas_function$ ! MEASUREMENT FUNCTION
1760  OUTPUT Hp4284a;"CORR:LOAD:TYPE "&Corr_function$ ! LOAD CORRECTION FUNCTION
1770  OUTPUT Hp4284a;"INIT:CONT ON"           ! 4284A INITIALIZE
1780  OUTPUT Hp4284a;"TRIG:SOUR BUS"          ! TRIG MODE
1790  OUTPUT Hp4284a;"CORR:OPEN:STAT 1"       ! OPEN CORR ON
1800  OUTPUT Hp4284a;"CORR:SHOR:STAT 1"       ! SHORT CORR ON
1810  OUTPUT Hp4284a;"CORR:SPOT1:STAT ON"     ! SPOT FREQ ON
1820  OUTPUT Hp4284a;"CORR:SPOT1:FREQ "&VAL$(Freq)        ! SPOT FREQ
1830  IF Correct_sw=Open_short THEN OUTPUT Hp4284a;"CORR:LOAD:STAT OFF"
1840  IF Correct_sw=Open_short_load THEN OUTPUT Hp4284a;"CORR:LOAD:STAT ON"
1850  IF Correct_mode=Single THEN OUTPUT Hp4284a;"CORR:METH SING"
1860  IF Correct_mode=Multi THEN OUTPUT Hp4284a;"CORR:METH MULT"
1870  OUTPUT Hp4284a;"DISP:PAGE MEAS"         ! Display measrement page
1880  OUTPUT Hp3488a;"CMONZ"                  ! Card monitor
1890  IF Function=Measurement THEN Measurement ! Measurement or correction
1900                                          !<<Correction progress>>
1910  FOR Ch=1 TO Noch                        !<<Open correction>>
1920    Ch$=VAL$(Ch)                          !
1930    PRINT CHR$(12)                        ! Clear display
1950    PRINT "CH."&Ch$&" OPEN MEASUREMENT"   ! Print channel number
1960    PRINT "OPEN TEST TERMINALS OF CH"&Ch$ !
1970    PRINT "(1) START OPEN MEAS.   (2) SKIP CH."&Ch$&" ?"
1980    Work=0                                !
1990    INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
2000    IF Work<>1 AND Work<>2 THEN 1990      !
2010    IF Work=2 THEN Open_skip_ch           !
2020    OUTPUT Hp3488a;Close_ch$(Ch)          ! Close switch of HP3488A
2030    OUTPUT Hp4284a;"CORR:USE "&VAL$(Ch)   ! Select compen. ch. of HP4284A
2040    PRINT "OPEN MEASUREMENT IN PROGRESS"  !
2050    OUTPUT Hp4284a;"CORR:SPOT1:OPEN"      ! Exec. OPEN corr. data meas.
2060    OUTPUT Hp4284a;"STAT:OPER:EVEN?"      ! Check corr. meas. end
2070    ENTER Hp4284a;Event_reg               ! Enter event register
2080    IF BINAND(Event_reg,1)=0 THEN 2060    !
2090    OUTPUT Hp3488a;Open_ch$(Ch)           ! Open switch of HP3488A
2100  Open_skip_ch:                           !
2110    IF Correct_mode=Single THEN Short_meas !
2120  NEXT Ch                                 !
2130  Short_meas:                             !<<SHORT correction>>
2140  FOR Ch=1 TO Noch                        !
2150  Ch$=VAL$(Ch)                            !
2160  PRINT CHR$(12)                          ! Clear screen
2170    PRINT "CH."&Ch$&" SHORT MEASUREMENT"  ! Print channel number
2180    PRINT "SHORT TEST TERMINALS OF CH"&Ch$ !
2190    PRINT "(1) START SHORT MEAS.   (2) SKIP CH."&Ch$&" ?"
```

```
2200    Work=0                                     ! Clear Work
2210    INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
2220    IF Work<>1 AND Work<>2 THEN 2210           !
2230    IF Work=2 THEN Short_skip_ch               !
2240    OUTPUT Hp3488a;Close_ch$(Ch)               ! Close switch of HP3488A
2250    OUTPUT Hp4284a;"CORR:USE "&VAL$(Ch)        ! Select compen. ch. of HP4284A
2260    PRINT "SHORT MEASUREMENT IN PROGRESS"      !
2270    OUTPUT Hp4284a;"CORR:SPOT1:SHOR"           ! Exec. SHORT corr. data meas.
2280    OUTPUT Hp4284a;"STAT:OPER:EVEN?"           ! Check corr. meas. end
2290    ENTER Hp4284a;Event_reg                    ! Enter event register
2300    IF BINAND(Event_reg,1)=0 THEN 2280         !
2310    OUTPUT Hp3488a;Open_ch$(Ch)                ! Open switch of HP3488A
2320 Short_skip_ch:                                !
2330    IF Correct_mode=Single THEN Load_meas      !
2340  NEXT Ch                                      !
2350 Load_meas:                                    !<<LOAD correction>>
2360  IF Correct_sw=Open_short THEN Skip_load      ! OPEN/SHORT or OPEN/SHORT/LOAD
2370  FOR Ch=1 TO Noch                             !
2380  Ch$=VAL$(Ch)                                 !
2390  PRINT CHR$(12)                               ! Clear screen
2400    PRINT "CH."&Ch$&" LOAD MEASUREMENT"        ! Print channel number
2410    PRINT "CONNECT A STANDARD TO THE TEST TERMINALS OF CH"&Ch$
2420    PRINT "(1) START LOAD MEAS.   (2) SKIP CH."&Ch$&" ?"
2430    Work=0                                     !
2440    INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
2450    IF Work<>1 AND Work<>2 THEN 2440           !
2460    IF Work=2 THEN Load_skip_ch                !
2470    OUTPUT Hp3488a;Close_ch$(Ch)               ! Close switch of HP3488A
2480    OUTPUT Hp4284a;"CORR:USE "&VAL$(Ch)        ! Select compen. ch. of HP4284A
2490    PRINT "LOAD MEASUREMENT IN PROGRESS"       !
2500    OUTPUT Hp4284a;"CORR:SPOT1:LOAD"           ! Exec. LOAD corr. data meas.
2510    OUTPUT Hp4284a;"STAT:OPER:EVEN?"           ! Check corr. meas. end
2520    ENTER Hp4284a;Event_reg                    ! Enter event register
2530    IF BINAND(Event_reg,1)=0 THEN 2510         !
2540    OUTPUT Hp3488a;Open_ch$(Ch)                ! Open switch of HP3488A
2550    PRINT CHR$(12)                             ! Clear screen
2560    PRINT "CH.";VAL$(Ch);"  STANDARD VALUE SETTING"
2570    PRINT "INPUT STANDARD VALUE OF A"          !
2580    INPUT "TYPE VALUE AND PRESS RETURN KEY",Ref_a
2590    PRINT "INPUT STANDARD VALUE OF B"          !
2600    INPUT "TYPE VALUE AND PRESS RETURN KEY",Ref_b
2610    PRINT "STANDARD VALUE A=";Ref_a;",  B=";Ref_b;"  OK?  (1) YES   (2) NO"
2620    INPUT "TYPE NUMBER AND PRESS RETURN KEY",Work
2630    IF Work<>1 AND Work<>2 THEN 2620           !
2640    IF Work=2 THEN 2550                        !
2650    OUTPUT Hp4284a;"CORR:SPOT1:LOAD:STAN "&VAL$(Ref_a)&","&VAL$(Ref_b)
2660 Load_skip_ch:                                 !
2670    IF Correct_mode=Single THEN End_correction
2680    NEXT Ch                                    !
2690 Skip_load:                                    !
2700    OUTPUT Hp3488a;Open_ch$(Ch)                ! Open switch of HP3488A
2710 End_correction:                               !
2720  PRINT CHR$(12)                               ! Clear screen
2730  PRINT "CORRECTION IS COMPLETED."             !
2740  Work=0                                       !
2750  INPUT "DO YOU WANT TO CONTINUE TO MEASURE?   (1) YES   (2) NO ",Work
2760  IF Work=1 THEN Measurement                   !
2770  IF Work=2 THEN STOP                          !
2780  IF Work<>1 AND Work<>2 THEN 2750             !
2790                                               !
```

```
2800 Measurement:                                  !<<Measurement progress>>
2810  PRINT CHR$(12)                               ! Clear screen
2820  PRINT "CORR: "&Correct_sw$&"   MODE:  "&Correct_mode$
2840  FOR Ch=1 TO Noch                             ! Scanning loop top
2850    OUTPUT Hp3488a;Close_ch$(Ch)               ! Close switch of HP3488A
2860    OUTPUT Hp4284a;"CORR:USE "&VAL$(Ch)        ! Setting a used channel
2870    OUTPUT Hp4284a;"INIT"                      ! Initialize HP 4284A
2880    OUTPUT Hp4284a;"*TRG"                      ! Triggering
2890    ENTER Hp4284a;Work$                        ! Enter measurement data
2900    A(Ch)=VAL(Work$[1,12])                     !
2910    B(Ch)=VAL(Work$[14,25])                    !
2920    PRINT "CH.";Ch,A$&"=";A(Ch),B$&"=";B(Ch)! Print data
2930    A_sum=A_sum+A(Ch)                          ! Summation of A
2940    B_sum=B_sum+B(Ch)                          ! Summation of B
2950    OUTPUT Hp3488a;Open_ch$(Ch)                ! Open switch of HP3488A
2960  NEXT Ch                                      ! Scanning loop end
2970                                               ! End of measurement
2980  A_ave=A_sum/Noch                             ! Calculate average of A
2990  B_ave=B_sum/Noch                             ! Calculate average of B
3000  FOR Ch=1 TO Noch                             !
3010    A_sx=(A_ave-A(Ch))^2+A_sx                  !
3020    B_sx=(B_ave-B(Ch))^2+B_sx                  !
3030  NEXT Ch                                      !
3040  A_sd=SQR(A_sx/Noch)                          ! Calculate S.D. of A
3050  B_sd=SQR(B_sx/Noch)                          ! Calculate S.D. of B
3060  PRINT                                        !
3070  PRINT A$&" AVE.=";A_ave,B$&" AVE.=";B_ave ! Print average of A & B
3080  PRINT A$&" S.D.=";A_sd,B$&" S.D.=";B_sd   ! Print S.D. of A & B
3090  STOP                                         !
3100  END                                          !
```

HEWLETT PACKARD

For more information, call your local HP sales office listed in the telephone directory white pages. Ask for the Electronic Instrument Department, or write to Hewlett-Packard **U.S.A.** - P.O. Box 10301, Palo Alto, CA 94303-0890 **Europe** - Hewlett-Packard S.A., P.O. Box 529, 1180 AM Amstelveen, The Netherlands. **Canada** - 6877 Goreway Drive, Mississauga, L4V 1M8, Ontario **Japan** - Yokogawa-Hewlett-Packard Ltd., 3-29-21, Takaido-Higashi, Suginami-ku, Tokyo 168 **Far East** - Hewlett-Packard Asia Headquarters, 47/F China Resources Building, 26 Harbour Road, Wanchai Hong Kong **Australasia** - Hewlett-Packard Australia Ltd., 31-41 Joseph Street, Blackburn, Victoria 3130 Australia **Latin America** - Hewlett-Packard Latin America Headquarters, 3495 Deer Creek Rd., Palo Alto, CA 94304 For all other areas please write to Hewlett-Packard Intercontinental Headquarters, 3495 Deer Creek Rd., Palo Alto, CA 94304

Printed in Japan | Sept., 1988 | 5950-2975